Eserciziario Pascal versione 1.0 Novembre 2004



NOTA: il turbo pascal (con alcuni PC) è affetto da un problema noto che manda in errore i programmi durante l’esecuzione se viene usata la unit crt; scrivete un programma di prova che usi la unit crt (anche vuoto che non fa niente!) e mandatelo in esecuzione: 

· se otterrete il messaggio di errore 'run time error 200', allora avete bisogno della correzione che prevede semplicemente di indicare *prima* della unit crt la unit newdelay; sul mio sito www.camuso.it troverete la newdelay.tpu (per chi usa turbo pascal 7) e la newdelay.pas (da compilare con altre versioni del Turbo Pascal) per ottenere sempre newdelay.tpu;    

la newdelay.tpu può essere copiata ogni volta (scomodo) nella cartella che contiene il sorgente pascal che la usa; oppure (molto meglio) copiata una volta per tutte in una cartella a vostra scelta, il cui percorso deve essere però indicato con il comando del menu Turbo Pascal  'Options/directories': nella prima casella scrivete (o aggiungete separando con un punto e virgola) il percorso di questa cartella;

è normale che in partenza il programma stampi 'crt ini bug trapped!': è la conferma che la newdelay sta funzionando! E' sufficiente iniziare il proprio programma con clrscr per cancellare lo schermo ...

attenzione: la newdelay deve essere indicata *prima* di crt

esiste anche una VIDEOLEZIONE preparata da me e che può essere trovata sempre sul mio sito (il titolo è: ambiente di lavoro turbo pascal)



· se non otterrete alcun messaggio di errore ignorate la procedura precedente ed eliminate semplicemente dagli esempi la unit newdelay che appariranno dunque così:

     program xxxxx;
     uses crt;
     …  ecc.   …

    ATTENZIONE: se non avete bisogno della newdelay e non eliminerete dagli esempi il 
    riferimento ad essa, otterrete il messaggio di errore ‘file not found’ (perché non viene trovato il 
    file newdelay.tpu


STRUTTURA SELETTIVA

SEL1. difficoltà: bassa Inserita un’età dire se siamo in presenza di un minorenne o maggiorenne.

Program maggiorenni1;

uses

  newdelay, crt;

var

  eta: integer;   (* variabile in cui verra' memorizzata l'eta' *)

                  (* NOTA: non usare mai le lettere accentate per i

                           nomi delle variabili ! *)
begin

  clrscr;  (* cancello lo schermo *)

  write('Quanti hanni hai? -> ');

  readln(eta);

  if eta>=18 then

    writeln('OK, vedo che sei maggiorenne ...')

  else

    writeln('Sei ancora un poppante, torna tra ', 18 - eta, 'anni !!');

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

(* problemi aperti: cosa succede se chi usa il programma sbaglia ed inserisce

   un eta' negativa o troppo grande per essere vera ?

   modifica il programma per controllare queste possibilita' !!

   trovi la soluzione nel sorgente 'maggior2.pas' *)
end.
SEL2. difficoltà: media Inserita un'età', dire se siamo in presenza di un maggiorenne o di un

minorenne; controllare anche eventuali errori di inserimento da parte dell'utente
program maggiorenni2;

uses  newdelay, crt;

var

  eta: integer;   (* variabile in cui verra' memorizzata l'eta' *)

                  (* NOTA: non usare mai le lettere accentate per i

                           nomi delle variabili ! *)

begin

  clrscr;  (* cancello lo schermo *)

(* rispetto alla soluzione dell'esercizio maggior.pas indichiamo nella richiesta dell'eta' anche l'intervallo accettato: da 0 anni (neonato) a 120 *)

  write('Quanti hanni hai (0-120) ? -> ');

  readln(eta);

(* SOLUZIONE 1 *)

  (* controlliamo per errori: eta' negative o superiori a 120 *)

  writeln('SOLUZIONE 1');

  if ( eta<0 ) or ( eta>120 ) then

    writeln('ERRORE: l''eta'' deve essere compresa tra 0 e 120 !')

  else

    if eta>=18 then

      writeln('OK, vedo che sei maggiorenne ...')

    else

      writeln('Sei ancora un poppante, torna tra ', 18 - eta, 'anni !!');

(* commenti: notate le parentesi OBBLIGATORIE quando in un if c'e' più
   di una condizione; notate come sia possibile iniziare una parte ELSE

   subito con un altro test; questo tipo di struttura

         if ... then

           ...

         else

           if ... then

             ...

           else

             if ... then

             ecc. ecc.

   e' detta con if in cascata e permette di verificare in successione una

   sequenza di condizioni

   La soluzione precedente non consente, in caso di errore, di capire

   se lo sbaglio e' relativo ad un'età' minore di zero o ad un'età' troppo

   grande perchè le due condizioni sono testate insieme. Vediamo come

   ottenere un controllo ancora piu' accurato rispondendo con messaggi

   di errore differenziati e piu' accurati*)

(* SOLUZIONE 2 *)

  writeln('------------------------------');

  writeln('SOLUZIONE 2');

  if ( eta<0 ) then

    writeln('ERRORE: l''eta'' non puo''essere negativa!')

  else

    if ( eta>120 ) then

      writeln('ERRORE: l''eta'' non puo''essere maggiore di 120!')

    else

      if eta>=18 then

        writeln('OK, vedo che sei maggiorenne ...')

      else

        writeln('Sei ancora un poppante, torna tra ', 18 - eta, ' anni !!');

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

SEL3. difficoltà: bassa Inserito un numero, dire se e' pari o dispari 
program paridispari1;

uses newdelay, crt;

var

  numero: integer;   (* variabile in cui verra' memorizzata il numero *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Inserisci un numero e ti diro'' se e'' pari o dispari -> ');

  readln(numero);

  if numero mod 2 = 0 then

    writeln('e'' pari')

  else

    writeln('e'' dispari');

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.
SEL4. difficoltà: bassa Inserito un carattere, dire se e' una vocale od una consonante. 

program consonanteVocale;

uses

  newdelay, crt;

var

  c: char;   (* variabile in cui verra' memorizzata il singolo carattere *)

(* commento: il tipo char può memorizzare un solo carattere; occupa la

   meta' dello spazio di una stringa lunga un solo carattere perché con le

   stringhe un byte supplementare serve a memorizzare la lunghezza della

   stringa stessa; questo non e' invece ovviamente necessario per una

   variabile carattere, la cui lunghezza e'' sempre 1 *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Inserisci un carattere -> ');

  readln(c);

(* PRIMA SOLUZIONE *)

  writeln('PRIMA SOLUZIONE');

  if ( c='a' ) or ( c='e' ) or ( c='i' ) or ( c='o' ) or ( c='u' ) then

    writeln('e'' una vocale')

  else

    writeln('e'' una consonante');

  writeln('--------------------------');

  writeln;

(* commento: questa soluzione non tiene conto delle vocali in maiuscolo *)

(* SECONDA SOLUZIONE *)

  writeln('SECONDA SOLUZIONE');

  if ( c='a' ) or ( c='e' ) or ( c='i' ) or ( c='o' ) or ( c='u' ) or

     ( c='A' ) or ( c='E' ) or ( c='I' ) or ( c='O' ) or ( c='U' )

  then

    writeln('e'' una vocale')

  else

    writeln('e'' una consonante');

  writeln('--------------------------');

  writeln;

(* commento: provate con la A: la prima soluzione non fornisce un risultato

   corretto, la seconda si'; pero'' ci sono molti test ... ecco una soluzione

   piu'' efficiente *)

(* TERZA SOLUZIONE *)

  writeln('TERZA SOLUZIONE');

  (* prima trasformiamo il carattere in maiuscolo e poi facciamo i test

     solo con le maiuscole; upcase e' una funzione che restituisce il

     maiuscolo del carattere fornito come parametro tra parentesi;

     se il carattere era gia' in maiuscolo, rimane in maiuscolo *)

  c := upcase( c );

  if ( c='A' ) or ( c='E' ) or ( c='I' ) or ( c='O' ) or ( c='U' ) then

    writeln('e'' una vocale')

  else

    writeln('e'' una consonante');

  writeln('--------------------------');

  writeln;

(* QUARTA SOLUZIONE *)

  writeln('QUARTA SOLUZIONE');

  (* soluzione con if in cascata, poco leggibile e lunga *)

  c := upcase( c );

  if ( c<>'A' ) then

    if ( c<>'E' ) then

      if ( c<>'I' ) then

        if ( c<>'O' ) then

          if ( c<>'U' ) then

            writeln('e'' una consonante')

          else

            writeln('e'' una vocale (U)')

        else

          writeln('e'' una vocale (O)')

      else

        writeln('e'' una vocale (I)')

    else

      writeln('e'' una vocale (E)')

  else

    writeln('e'' una vocale (A)');

(* commenti: questa soluzione, piuttosto pesante, consente pero' di

   intraprendere azioni differenziate a seconda delle vocali incontrate;

   i blocchi ELSE sono tutti necessari! diversamente inserendo quella vocale

   non verrebbe stampato nulla! Prova a togliere l'ultimo blocco ed inserisci

   la A ... *)

  writeln('--------------------------');

  writeln;

(* QUINTA SOLUZIONE *)

  writeln('QUINTA SOLUZIONE');

  (* la piu' efficiente, sfruttando una delle funzioni disponibili

     per le stringhe: la funzione pos restituisce la prima posizione di una

     stringa all'interno di un altra; se non la trova restituisce zero;

     ad esempio pos('po','topolino') restituisce 3 perche' la stringa 'po'

     e' stata trovata a partire dalla posizione 3 nella stringa 'topolino' *)

  if pos(c,'aAeEiIoOuU') <> 0 then  (* c trovata da qualche parte *)

    writeln('e'' una vocale')

  else

    writeln('e'' una consonante');

  writeln('--------------------------');

  writeln;

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

SEL5. difficoltà: bassa Inseriti A, B e C dire se B e' compreso tra A e C; in pratica si controlla se B appartiene all'intervallo [A,C]
program intervallo;

uses

  newdelay, crt;

var

  a, b, c: real;

  (* variabili in cui verranno memorizzati i numeri; in pratica inizio  *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Inserisci l''estremo sinistro dell''intervallo -> ');

  readln(a);

  write('Inserisci l''estremo destro dell''intervallo -> ');

  readln(c);

  (* se i numeri sono stati inseriti nell'ordine sbagliato, stop *)

  if a>c then

    writeln('Intervallo impossibile.')

  else

  begin

    write('Inserisci un numero e ti diro'' se appartiene all''intervallo -> ');

    readln(b);

    if ( b>=a ) and ( b<=c ) then

      writeln('Appartiene')

    else

      writeln('Non appartiene')

  end;

(* commenti: si e' deciso di comprendere gli estremi (>= e <=) nei controlli;

   notate le parentesi obbligatorie in presenza di piu' di una condizione;

   convincetevi che OR invece di and sarebbe stato un errore: un numero,

   per essere nell'intervallo [a,c] deve essere CONTEMPORANEAMENTE piu'

   grande (al massimo uguale) di a e piu' piccolo (al massimo uguale) di c;

   usando l'OR la condizione sarebbe stata vera anche quando UNA SOLA delle

   condizioni e' vera; ad esempio se l'intervallo fosse [5, 11], ed il numero

   da verificare fosse 13, con  ( b>=a ) or ( b<=c ) andremmo in pratica a

   verificare ( 13>=5 ) or ( 13<=11 ) che risulterebbe vera grazie 13>=5 anche

   se e' falso che 13<=11 (e quindi il punto non appartiene all'intervallo!

   l'or (ma con diverse condizioni) e' il connettivo logico giusto quando

   si vuole controllare che un punto sia ESTERNO ad un intervallo:

    if ( b<a ) or ( b>c ) then

      writeln('Non appartiene')

    else

      writeln('Appartiene')

   infatti un punto NON appartiene se e' piu' piccolo dell'estreo inferiore

   o piu' grande dell'estremo superiore

*)

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.
SEL6. difficoltà: bassa Inserite due misure, una in pollici e l'altra in centimetri, dire qual è la maggiore
program centimetriPollici;

uses

  newdelay, crt;

var

  cm: real;   (* variabile in cui verra' memorizzate la misura in centimetri *)

  po: real;   (* variabile in cui verra' memorizzate la misura in pollici *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Inserisci la misura in centimetri -> ');

  readln(cm);

  write('Inserisci la misura in pollici -> ');

  readln(po);

  (* ricordando che 1 pollice=2.54 cm ... *)

  if po * 2.54 > cm then

    writeln('La misura in pollici e'' maggiore (', po*2.54:4:2, 'cm)')

  else

    if po * 2.54 < cm then

      writeln('La misura in centimetri e'' maggiore')

    else

      writeln('Le due misure sono equivalenti');

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.
SEL 7. difficoltà: bassa Inserite le misure dei lati di 2 rettangoli dire quale dei due ha la superficie maggiore
program confrontaAree;

uses

  newdelay, crt;

var

  b1, b2: real;   (* variabile in cui verranno memorizzate le due basi *)

  h1, h2: real;   (* variabile in cui verranno memorizzate le due altezze *)

  a1, a2: real;   (* variabile in cui verranno memorizzate le due aree *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Inserisci la base del primo rettangolo -> ');

  readln(b1);

  write('Inserisci l''altezza del primo rettangolo -> ');

  readln(h1);

  write('Inserisci la base del secondo rettangolo -> ');

  readln(b2);

write('Inserisci l''altezza del secondo rettangolo -> ');

  readln(h2);

(* PRIMA SOLUZIONE *)

  writeln('PRIMA SOLUZIONE');

  (* se non interessa memorizzare i valori delle due aree si puo' procedere

     al confronto diretto delle rispettive formule di calcolo *)

  if b1*h1 > b2*h2 then

    writeln('Il primo rettangolo ha la superficie maggiore')

  else

    if b1*h1 < b2*h2 then

      writeln('Il secondo rettangolo ha la superficie maggiore')

    else

      writeln('I due rettangoli hanno la stessa superficie');

  writeln('--------------------------');

  writeln;

(* SECONDA SOLUZIONE (migliore) *)

  writeln('SECONDA SOLUZIONE');

  (* calcoliamo prima le due superfici e salviamo il risultato in a1 e a2 *)

  a1 := b1*h1;

  a2 := b2*h2;

  (* poi usiamo a1 e a2 per i confronti *)

  if a1>a2 then

    writeln('Il primo rettangolo ha la superficie maggiore')

  else

    if a1<a2 then

      writeln('Il secondo rettangolo ha la superficie maggiore')

    else

      writeln('I due rettangoli hanno la stessa superficie');

  writeln('--------------------------');

  writeln;

(* commenti: questa seconda soluzione usa due variabili in più però ha il

   grosso pregio di non far ripetere il calcolo delle aree; inoltre

   permetterebbe di usare i valori delle aree anche in una ipotetica

   continuazione del programma; rende anche più semplice comprendere

   quello che si sta facendo *)

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.
SEL 8. difficoltà: media Letto un carattere, dire se corrisponde ad una lettera maiuscola e se sì
   dire se e' una consonante od una vocale. 
program maiuscola;

var

  c: char;

begin

  write('Inserire un carattere -> ');

  readln(c);

(* PRIMA SOLUZIONE: SENZA CASE *)

(* se il codice ASCII del carattere letto e' compreso tra quelli di A e Z

   allora si tratta di una maiuscola ... *)

writeln('SENZA ''CASE'' ...');

if ( ord(c)>=ord('A') ) and ( ord(c)<=ord('Z') ) then

begin

  writeln('Hai inserito una maiuscola');

  (* vediamo ora se c e' uguale ad una delle vocali *)

  if (c='A') or (c='E') or (c='I') or  (c='O') or (c='U') then

    writeln('... ed e'' una vocale')

  else

    writeln('... ed e'' una consonante')

end

else

  writeln('Non si tratta di una maiuscola');

(* POI SOLUZIONE CON CASE *)

writeln('CON ''CASE'' ...');

(* il case piu' esterno ne contiene interamente un altro in un suo caso *)

case ord(c) of

  ord('A')..ord('Z'):  (* .. significa dal valore .. al valore *)

    begin

      writeln('Hai inserito una maiuscola');

      (* questo case e' contenuto nel primo caso di quello piu' esterno *)

      case c of

      'A','E','I','O','U': writeln('... ed e'' una vocale');

      else

        writeln('... ed e'' una consonante')

      end;

    end;

else

  writeln('Non si tratta di una maiuscola');

end;

readln;

end.

Ricordo che la funzione ord(c), dove c è un carattere, restituisce il codice ASCII di quest’ultimo. Controllare che ord(c) sia compreso tra ord(‘A’) e ord(‘Z’) significa controllare che il codice ASCII del carattere c sia compreso tra quello della prima e dell’ultima lettera maiuscola e, in definitiva, che c contenga una lettere maiuscola.

SEL 9. difficoltà: alta Verificare se una data inserita da tastiera e' corretta

program dataCorretta;

var gg,mm,aa: integer; (* gg=giorno, mm=mese, aa=anno *)

    ok_data: boolean;

begin

  write('Inserire il giorno: ');readln(gg);

  write('Inserire il mese: ');readln(mm);

  write('Inserire l''anno: ');readln(aa);

  (* PRIMA SOLUZIONE SENZA CASE *)

  ok_data:=false; (* se data supera controlli ok_data verra' messo a true *)

  if (gg>0) and (gg<32) and (mm>0) and (mm<13) and (aa>0) then

    if mm=2 then  (* febbraio *)

    begin

      if aa mod 4 = 0 then (* per semplicita': bisestili se multipli di 4 *)

      begin

        if gg<=29 then

          ok_data:=true

      end

      else (* non bisestile *)

        if gg<=28 then

          ok_data:=true

    end

    else (* mese diverso da febbraio ...  *)

      if (mm=1) or (mm=3)or (mm=5) or (mm=7) or

         (mm=8) or (mm=10) or (mm=12)  then (* mese con 31 giorni *)

      begin

        if gg<=31 then

          ok_data:=true

      end

      else (* mese con 30 giorni *)

        if gg<=30 then

          ok_data:=true;

  writeln('Senza case:');

  if ok_data then

    writeln('Data corretta ... ')

  else

    writeln('Data errata !!');

  readln;

  (* POI SOLUZIONE CON CASE *)

  ok_data:=false;

  if (gg>0) and (gg<32) and (mm>0) and (mm<13) and (aa>0) then

    case mm of

      2: if aa mod 4 = 0 then

         begin

           if gg<=29 then

             ok_data:=true

         end

         else (* non bisestile *)

           if gg<=28 then

             ok_data:=true;

      1,3,5,7,8,10,12: if gg<=31 then

                         ok_data:=true;

    else

      if gg<=30 then

        ok_data:=true;

    end;

  writeln('Con il case:');

  if ok_data then

    writeln('Data corretta ... ')

  else

    writeln('Data errata !!');

  readln;

end.
SEL 10. difficoltà: media   Disegna il diagramma di flusso e scrivi un programma che, letti tre numeri,  li metta in ordine crescente
program ordina_tre;

var a,b,c,temp: integer;

begin

  write('Inserire il primo numero -> ');

  readln(a);

  write('Inserire il secondo numero -> ');

  readln(b);

  write('Inserire il terzo numero -> ');

  readln(c);

  (* e' necessario assicurarsi che tre numeri siano in ordine *)


  (* prima ordina a e b tra loro *)

  if b<a then

  begin

    temp:=a;

    a:=b;

    b:=temp

  end;

  (* poi metti c al posto giusto ... *)

  if c<a then

  begin

    temp:=a;

    a:=c;

    c:=b;

    b:=temp

  end

  else

    if c<b then

    begin

      temp:=b;

      b:=c;

      c:=temp

    end;

  writeln (a,' ',b,' ',c,' premere INVIO per continuare');

  readln;

end.

SEL 11. difficoltà: media  Letti tre numeri da tastiera, determinare se possono appartenere ad una progressione aritmetica. TRACCIA: in una progressione aritmetica la differenza tra due numeri

consecutivi e' costante. Ad esempio: 1  4   7  10  13  16 ecc. è la progressione in cui ogni numero si ottiene sommando 3 al precedente. La differenza tra due numeri consecutivi qualsiasi è pari a 3.
program progressione;

var a,b,c,temp: integer;

begin

  write('Inserire il primo numero -> ');

  readln(a);

  write('Inserire il secondo numero -> ');

  readln(b);

  write('Inserire il terzo numero -> ');

  readln(c);

  (* e' necessario prima assicurarsi che i tre numeri siano in ordine *)

  (* prima ordina a e b tra loro *)

  if b<a then

  begin

    temp:=a;

    a:=b;

    b:=temp

  end;

  (* poi metti c al posto giusto ... *)

  if c<a then

  begin

    temp:=a;

    a:=c;

    c:=b;

    b:=temp

  end

  else

    if c<b then

    begin

      temp:=b;

      b:=c;

      c:=temp

    end;

  if (b-a) = (c-b) then

    writeln('Sono in progressione aritmetica')

  else

    writeln('NON sono in progressione aritmetica');

  writeln ('premere INVIO per continuare');

  readln;

end.

STRUTTURA ITERATIVA
ITE1. difficoltà: bassa stampa dei primi N numeri naturali, con N letto da tastiera
program numeri;

uses

  newdelay, crt;

var

  i: integer;   (* contatore ciclo *)

  n: integer;   (* qui viene memorizzato il numero letto da tastiera *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Fino a che numero devo arrivare? -> ');

  readln( n );

  for i := 1 to n do

    writeln( i );

(* commenti: e' ancora materia di disputa tra matematici se considerare lo

   zero un naturale; se lo si vuole considerare tale il codice diventa:

     for i := 0 to n-1 do

       writeln( i );

   infatti se chiedessimo i primi 5 naturali dovremmo ottenere: 0,1,2,3,4

   e sarebbe quindi necessario fermare il ciclo a 4, cioe' n-1

 *)

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE2. difficoltà: bassa stampa dei primi N numeri naturali, con N letto da tastiera; la stampa deve avvenire dal numero piu' grande al piu' piccolo
program numeri;

uses

  newdelay, crt;

var

  i: integer;   (* contatore ciclo *)

  n: integer;   (* qui viene memorizzato il numero letto da tastiera *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Fino a che numero devo arrivare? -> ');

  readln( n );

  for i := n downto 1 do   (* NOTATE IL DOWNTO INVECE DI TO !!!! *)

    writeln( i );

(* commenti: e' ancora materia di disputa tra matematici se considerare lo

   zero un naturale; se lo si vuole considerare tale il codice diventa:

     for i := n-1 to 0 do

       writeln( i );

   infatti se chiedessimo i primi 5 naturali dovremmo ottenere: 4,3,2,1,0

   e sarebbe quindi necessario iniziare il ciclo da 4, cioe' n-1

 *)

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE3. difficoltà: bassa stampa dei primi N numeri naturali, con N letto da tastiera; a fianco di ciascun numero indicare se e' pari o dispari
program numeri;

uses

  newdelay, crt;

var

  i: integer;   (* contatore ciclo *)

  n: integer;   (* qui viene memorizzato il numero letto da tastiera *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Fino a che numero devo arrivare? -> ');

  readln( n );

(* ricordo che l'operatore MOD calcola il resto della divisione tra il

   numero che viene messo alla sua sinistra e quello che viene messo alla

   sua destra; ad esempio 7 mod 3 calcola il resto della divisione tra

   7 e 3, quindi 1; 13 mod 8 -> 5; 19 mod 5 -> 4;

   quando il primo numero e' multiplo del secondo il resto e' zero:

   4 mod 2 -> 0; 15 mod 3 -> 0; 15 mod 5 -> 0;

*)

  for i := 1 to n do

  begin

    write( i );

    (* i numeri pari divisi per due danno resto 0 *)

    if i mod 2 = 0 then

      writeln(' numero pari')

    else

      writeln(' numero dispari');

  end;

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE4. difficoltà: bassa Stampa dei numeri dispari minori o uguali a N, con N letto da tastiera
program dispari;

uses

  newdelay, crt;

var

  i: integer;   (* contatore ciclo *)

  n: integer;   (* qui viene memorizzato il numero letto da tastiera *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Fino a che numero devo arrivare? -> ');

  readln( n );

(* ricordo che l'operatore MOD calcola il resto della divisione tra il

   numero che viene messo alla sua sinistra e quello che viene messo alla

   sua destra; ad esempio 7 mod 3 calcola il resto della divisione tra

   7 e 3, quindi 1; 13 mod 8 -> 5; 19 mod 5 -> 4;

   quando il primo numero e' multiplo del secondo il resto e' zero:

   4 mod 2 -> 0; 15 mod 3 -> 0; 15 mod 5 -> 0;

*)

  for i := 1 to n do

  begin

    (* i numeri dispari divisi per due danno resto 1 *)

    if i mod 2 = 1 then

      writeln( i );

  end;

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE5. difficoltà: bassa Stampa dei primi N numeri naturali (N letto da tastiera) con a fianco

   i rispettivi quadrati e cubi

program potenze;

uses

  newdelay, crt;

var

  i: integer;   (* contatore ciclo *)

  n: integer;   (* qui viene memorizzato il numero letto da tastiera *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Fino a che numero devo arrivare? -> ');

  readln( n );

(* PRIMA SOLUZIONE *)

  writeln('PRIMA SOLUZIONE');

  for i := 1 to n do

    writeln( i, ' ', i*i, ' ', i*i*i );

  writeln('--------------------------');

  writeln;

(* SECONDA SOLUZIONE (sfrutta l'operatore sqr che calcola il quadrato *)

  writeln('SECONDA SOLUZIONE');

  for i := 1 to n do

    writeln( i, ' ', sqr(i), ' ', sqr(i)*i );

  writeln('--------------------------');

  writeln;

(*  TERZA SOLUZIONE

    allinea i numeri usando gotoxy(colonna, riga) per posizionare il

    cursore alla colonna/riga dello schermo desiderata;

    whereY e' una funzione che restituisce il numero di riga dove si

    trova il cursore; quindi gotoxy(7, whereY) significa colonna 7,

    non muoverti dalla riga su cui ti trovi ...*)

  writeln('TERZA SOLUZIONE');

  for i := 1 to n do

  begin

    write( i );

    gotoxy(7,whereY); write( sqr(i) );

    gotoxy(14,whereY); writeln( sqr(i)*i )

  end;

  writeln('--------------------------');

  writeln;

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE6. difficoltà: bassa Stampa dei numeri interi relativi da -N a +N, con N letto da tastiera
program numeri;

uses

  newdelay, crt;

var

  i: integer;   (* contatore ciclo *)

  n: integer;   (* qui viene memorizzato il numero letto da tastiera *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Fino a che numero devo arrivare? -> ');

  readln( n );

  for i := -n to n do

    writeln( i );

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE7. difficoltà: bassa Stampa dei numeri da INIZIO a FINE, con INIZIO e FINE letti da tastiera.

   Controllare che INIZIO sia <= FINE *)

program numeri;

uses

  newdelay, crt;

var

  i: integer;   (* contatore ciclo *)

  inizio, fine: integer;

  temp: integer;  (* variabile di appoggio per scambiare inizio e fine

                     se inseriti nell'ordine sbagliato *)

begin

  clrscr;  (* cancello lo schermo *)

  write('Numero di partenza -> ');

  readln( inizio );

  write('Numero finale -> ');

  readln( fine );

(* PRIMA SOLUZIONE *)

  writeln('PRIMA SOLUZIONE');

  if inizio<=fine then

    for i := inizio to fine do

      writeln( i )

  else

    for i := fine to inizio do

      writeln( i );

  writeln('--------------------------');

  writeln;

(* commenti: se l'utente inserisce inizio e fine nell'ordine sbagliato

   (inizio>fine) si esegue un ciclo con gli estremi invertiti; *)

(* SECONDA SOLUZIONE *)

  writeln('SECONDA SOLUZIONE');

  (* se inizio e fine sono invertiti, li scambio *)

  if inizio>fine then

  begin

    temp:=inizio;

    inizio:=fine;

    fine:=temp

  end;

  (* ora inizio e fine sono senza dubbio nell'ordine giusto ... *)

  for i := inizio to fine do

    writeln( i );

  writeln('--------------------------');

  writeln;

(* commenti: se l'utente inserisce inizio e fine nell'ordine sbagliato

   (inizio>fine) si esegue un ciclo con gli estremi invertiti; *)

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE8. difficoltà: bassa generazione di numeri casuali
(* random.pas

Con questo esercizio imparerete:

  - l'utilizzo della funzione random per generare numeri in modo casuale

    (utile per simulazioni, giochi ecc.

  - l'utilizzo della funzione randomize per avere sequenze di numeri casuali

    sempre diverse

*)

program prova_random;

uses newdelay crt;  

var x,i,num: integer; (* non hanno un significato particolare *)
begin

  clrscr;

(* ESPERIENZA 1 *)

(* random(N) restituisce un numero casuale compreso tra 0 e N-1 *)

  x:=random(5); (* x diventa un numero compreso tra 0 e 4 *)

  writeln('Esperienza 1: ', x);

  (* possiamo usare random direttamente, senza memorizzare il risultato in x *)

  writeln('Esperienza 1: ',random(9));

  readln;

(* ESPERIENZA 2 *)

(* generiamo una sequenza di 10 numeri compresi tra 0 e 15 *)

  for i:=1 to 10 do

    writeln('Esperienza 2: ', random(16));

  readln;

(* NOTA BENE: tutte le volte che si fa ripartire il programma, random genera

   la stessa sequenza di numeri (osservatelo con l'esperienza n. 2 facendo

   partire il programma piu' volte).

   Se si vuole iniziare ogni volta con una sequenza diversa, usare il comando

   randomize: esso cambia la sequenza basandosi sul tempo scandito

   dall'orologio interno del computer  *)

(* ESPERIENZA 3 *)

(* riscrivete il ciclo dell'esperienza due facendolo precedere dal comando

   randomize; fate ripartire piu' volte il programma e notate come la sequenza

   dell'esperienza 3 cambi tutte le volte *)

  randomize;

  for i:=1 to 4 do

    writeln('Esperienza 3: ', random(16));

  readln;

(* ESPERIENZA 4 *)

(* Invece di un numero possiamo usare come parametro per la random una

   variabile intera *)

  write ('Dimmi un numero: ');

  readln(num);

  writeln('Ora genero una sequenza di 10 numeri tra 0 e ',num - 1);

  for i:=1 to 10 do

    writeln('Esperienza 4: ', random(num));

readln;

(* PROVA DA SOLO ...

   1. Cosa accade indicando 1 come parametro della random ?

   2. Cosa accade indicando 0 come parametro della random ?

   3. Cosa accade indicando un numero negativo come parametro della random ?

   4. Cosa accade indicando un numero reale come parametro della random ?

   5. Cosa accade indicando una variabile integer che contiene un valore

      negativo come parametro della random?

*)

(* ESERCIZI SUGGERITI

ES. 1: trovate un modo per generare numeri tra 1 e N;

---------------------------------------------------------------------------

ES. 2: sfruttando quanto scoperto nell'esercizio n. 1, generate una colonna

       di risultati di una schedina del totocalcio;

---------------------------------------------------------------------------

ES. 3: generate in modo casuale la sequenza dei nomi per le interrogazioni di una classe con solo quattro alunni che si chiamano Primo, Secondo, Terzo, e Quarto;

       SUGGERIMENTI: il problema sara' non estrarre piu' volte lo stesso nome;

       fate corrispondere ogni nome ad un numero; quando viene estratto quel

       numero, stampate il nome corrispondente e poi 'cancellate' quel nome

       copiando nella variabile che lo memorizza una 'X' al posto del nome; se

       generate un numero casuale che corrisponde ad una X (nome gia' estratto)

       dovete provare con un altro numero; per sapere quando fermarsi gestite

       un contatore che aumenterete solo quando non troverete una 'X'

-----------------------------------------------------------------------------

ES. 4: (da fare dopo che il prof. ha spiegato i vettori)

       vogliamo estendere l'esercizio n. 3 per classi fino ad un max di 30

       alunni;

       SUGGERIMENTI

         SOLUZIONE INGENUA (estrae molti piu' numeri del dovuto)

         caricate i nomi degli alunni in un vettore; a numero estratto

         corrisponde nome alunno in quella posizione del vettore; dopo

         aver estratto un alunno, al posto del suo nome nel vettore

         mettete una X; durante le estrazioni si ignorano le estrazioni

         che corrispondono a caselle del vettore contenenti X; per sapere

         quando fermarsi gestite un contatore che aumenterete solo quando

         estrarrete numeri 'buoni'

         SOLUZIONE OTTIMALE (si fanno esattamente N estrazioni se N sono i nomi)

         caricate i nomi degli alunni (diciamo 15, per esempio) in un vettore;

         la prima volta generate un numero tra 1 e 15: immaginiamo sia uscito

         il 7; dopo aver stampato il nome nella settima posizione del vettore,

         prendete il nome che sta in ultima posizione e lo copiate nella

         settima; in questo modo i nomi che rimangono da estrarre sono stati

         concentrati nelle prime 14 posizioni; estraete ora un un numero tra

         1 e 14: immaginiamo il 5; stampate il quinto nome e poi copiate

         l'ultimo (il quattordicesimo) nella quinta posizione; di nuovo, i nomi

         che rimangono da estrarre sono stati concentrati nelle prime 13

         posizioni; ripetere fino all'esaurimento dei nomi; NOTA: se durante

         il procedimento viene estratto proprio l'ultimo (il quindicesimo la

         prima volta o il quattordicesimo la seconda ecc.) non si deve

         ricopiare niente e si diminuisce solo di uno il  contatore.

-----------------------------------------------------------------------------

ES. 5:   sfruttando la tecnica della soluzione ottimale dell'esercizio 4

         simulare il gioco della tombola (semplificato):

         - con un vettore di 90 posizioni si rappresenta il sacchetto da cui

           si estraggono i numeri;

         - con un secondo vettore di 5 posizioni si rappresenta una cartella di

           gioco semplificata (si potranno fare solo ambi, terne, quaterne

           e cinquine); si immagina quindi che ci sia un solo giocatore con

           una sola cartella di gioco

         - simulare le estrazioni dal sacchetto dichiarando quando si fa ambo,

           terna ecc.

-----------------------------------------------------------------------------

*)

end.

ITE9. difficoltà: bassa   utilizza la random per estrarre numeri da 1 a N

program estrai_1_N;

uses newdelay, crt;  

var i,n: integer;

begin

  clrscr;

(* dato che random(N) estrae numeri tra 0 ed N-1, e' sufficiente aggiungere

   1 al risultato per avere numeri tra 0+1=1 e N-1+1=N

*)

  write('Che N vuoi usare? -> ');

  readln(n);

  (* estraiamo 20 numeri da 1 a n *)

  for i:=1 to 20 do

    writeln( random(n)+1 );

  (* CONCLUSIONI

     La formula che genera un numero casuale tra 1 e N e':

          RANDOM(N) + 1

  *)

  readln;

end.

ITE10. difficoltà: bassa utilizza la random per estrarre numeri tra A e B

program estrai_AB;

uses newdelay, crt;  

var i,a,b: integer;

begin

  clrscr;

(* random(N) estrae numeri tra 0 ed N-1; immaginiamo di voler estrarre numeri

   tra 5 e 12: e' sufficiente aggiungere a 5 un numero casuale tra 0 e 7

   intuitivamente, infatti, 5+0=5 e 5+7=12; in generale dovremo aggiungere

   al primo estremo (A) un numero casuale tra 0 e B-A

*)

  write('Che A vuoi usare? -> ');

  readln(a);

  write('Che B vuoi usare? -> ');

  readln(b);

  (* estraiamo 20 numeri da A a B *)

  (* NOTA: attenzione! random(B-A) estrarrebbe numeri tra 0 e (B-A) - 1

           e' quindi necessario indicare random (B-A+1) *)

  for i:=1 to 20 do

    writeln( A + random(B-A+1) );

  readln;

(* RIFLETTI ....  QUESTA PARTE E' MOOOOOLTO IMPORTANTE!

   E' piuttosto difficile ricordarsi la formula 'A + random(B-A+1)'.

   Sarebbe meglio avere a disposizione un comando facile da usare, come

   la sqrt(x) per il calcolO della radice quadrata, per intenderci. Potremmo

   chiamare il comando 'randomAB' e prevedere che tra parentesi vadano

   specificati A e B; un esempio di utilizzo del comando potrebbe essere:

      writeln( randomAB(10,20) ) per avere un numero casuale tra 10 e 20

   Questi 'comandi' che restituiscono un valore possono essere realizzati

   in Pascal (ed in tutti i linguaggi moderni) realizzando una 'funzione'

   di calcolo. Ecco come si fa:

   1. Dopo la sezione VAR del programma e prima del begin di inizio si

      dichiara una 'function' in questo modo:

      program ...

      var ...

      function randomAB(A,B: integer) : integer;

      begin

        randomAB := A + random(B-A+1)

      end;

      begin (* quello del programma *)
       ... writeln( randomAB(5,9) ); ...

      end.

      in pratica si tratta di un programmino separato all'interno di

      quello principale; tra il begin e l'end della function si scrivono

      le istruzioni che servono a calcolare il valore richiesto; queste

      istruzioni (una sola nel nostro esempio) possono usare le variabili

      indicate tra parentesi (A e B);

      il programmatore stabilisce il valore calcolato dalla function

      assegnando un valore al nome della function: randomAB:= ... ;

      naturalmente avremmo potuto inserire anche centinaia di istruzioni

      prima di questa: l'importante e' che la function termini

      assegnando un valore al suo nome, altrimenti chi la usera' non avrebbe

      nessun risultato utile!

      il ': integer' dopo le parentesi della prima riga della function indicano

      che il valore calcolato e' di tipo integer;

      Come si usa la funzione?

      Scrivendo il suo nome ed indicando tra parentesi i valori richiesti o

      le variabili che contengono i valori richiesti. La funzione viene prima

      calcolata ed e' come se al suo posto venisse sostituito il suo valore.

      Ecco alcuni esempi:

      writeln( randomAB(5,7) );

      x:=randomAB(5,y); si immagina che y sia una variabile integer

      if randomAB(x,y)>7 then ...  si immagina che x e y siano di tipo integer

   NOTA: una volta scritta la function puo' essere copiata/incollata in

         qualsiasi altro programma rimanendo perfettamente funzionante,

         essendo svincolata dal resto del programma, come un pezzo

         prefabbricato riutilizzabile in diversi tipi di costruzione

(* PROVA DA SOLO ...

   1. Cosa accade indicando come parametri due numeri uguali ?

   2. Cosa accade indicando A > B ?

   3. Prova a ricopiare la function della nota precedente ed usala

      nel ciclo for al posto della formula difficile da ricordare

   4. Scrivi una function che riceve un valore in lire e restituisce

      il corrispondente valore in euro e provala all'interno di questo

      programma

   5. Scrivi una function che riceve un numero tra 1 e 12 che rappresenta

      un mese dell'anno (1=gennaio, 2=febbraio ecc.) e restituisce il numero

      di giorni presente in quel mese; considerate febbraio sempre di 28 gg

   6. Scrivi una function che riceve un numero che rappresenta un anno e

      restituisce TRUE se l'anno e' bisestile, FALSE altrimenti; considerate

      un anno bisestile quando e' divisibile per 4 (regola semplificata)

   7. Scrivi una function che riceve un numero che rappresenta un mese

      ed un numero che rappresenta un anno e restituisce il numero dei

      giorni di quel mese di quell'anno; questa function usa quella

      dell'esercizio precedente per decidere, nel caso il mese sia 2,

      se i giorni sono 29 (bisestile) o 28

*)

(* CONCLUSIONI

     La formula che genera un numero casuale tra A e B e':

          A + RANDOM(B-A+1)

*)

end.

ITE11. difficoltà: bassa generazione di una colonna di una schedina del totocalcio

program colonna_totocalcio;

uses newdelay, crt; 

var i: integer;

    risultato: integer;

begin

  clrscr;

  randomize;

(* non potendo estrarre la X, useremo un numero al suo posto: il 3; generiamo

   allora 13 numeri da 1 a 3 ... *)

   for i:=1 to 13 do

   begin

     risultato:= random(3)+1;

     if  risultato = 3 then

       writeln('X')

     else

       writeln(risultato)   (* 1 o 2 *)

   end;

   readln;

(* proviamo con una schedina piu' realistica che privilegia gli 1 rispetto alle

   X e le X rispetto ai 2; si estrae un n. da 1 a 13 e se e' tra 1 e 6 e' come

   se fosse uscito l'1, tra 7 e 10 l'X e tra 11 e 13 il 2 *)

   for i:=1 to 13 do

   begin

     risultato:= random(13)+1;

     if  risultato <= 6 then

       writeln(1)

     else

     if risultato<=10 then

       writeln('X')

     else

       writeln(2)

   end;

   readln;

end.

ITE12. difficoltà: bassa far scrivere sullo schermo 100 volte la frase   'come programma mi sento un po' stupido ...' 
program messaggi;

uses

  newdelay, crt;

const

  QUANTE_VOLTE=100; (* numero dei messaggi che saranno stampati *)

(* commento: l'uso di una costante permette di cambiare molto

   facilmente il numero dei messaggi *)

var

  i: integer;   (* contatore ciclo *)

begin

  clrscr;  (* cancello lo schermo *)

  for i := 1 to QUANTE_VOLTE do

    writeln('come programma mi sento un po'' stupido ...');

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE13 difficoltà: bassa far scrivere sullo schermo 5 volte la frase   'come programma mi sento un po' stupido ...';  indicare all'inizio di ogni riga il suo numero progressivo (1, 2, 3 ...)

program messaggi;

uses

  newdelay, crt;

const

  QUANTE_VOLTE=5; (* numero dei messaggi che saranno stampati *)

(* commento: l'uso di una costante permette di cambiare molto

   facilmente il numero dei messaggi *)

var

  i: integer;   (* contatore ciclo *)

begin

  clrscr;  (* cancello lo schermo *)

(* PRIMA SOLUZIONE *)

  writeln('PRIMA SOLUZIONE');

  for i := 1 to QUANTE_VOLTE do

    writeln('N. riga: ', i ,' come programma mi sento un po'' stupido ...');

  writeln('--------------------------');

  writeln;

(* spesso nelle istruzioni di un ciclo for e' utile usare la variabile di

   controllo del ciclo stesso; in questo esempio, al momento di scrivere

   ogni riga la variabile di controllo i ha proprio il valore che serve

   stampare come numero progressivo di riga; infatti quando si stampa

   per la prima volta il messaggio il suo valore e' uno, quando si stampa

   il messaggio per la seconda volta il suo valore e' due ecc. *)

(* SECONDA SOLUZIONE *)

  writeln('SECONDA SOLUZIONE');

  for i := 1 to QUANTE_VOLTE do

  begin

    write('N. riga: ');(* non va a capo ... *)

    write(i);          (* non va a capo ... *)

    writeln(' come programma mi sento un po'' stupido ...');

  end;

  writeln('--------------------------');

  writeln;

(* commenti: la seconda soluzione e' equivalente alla prima; notate come sia

   INDISPENSABILE aggiungere un begin ed un end per racchiudere ed indicare

   in modo preciso l'inizio e la fine del blocco di istruzioni che deve

   essere ripetuto; se infatti avessimo scritto:

      for i := 1 to QUANTE_VOLTE do

        write('N. riga: ');

        write(i);

        writeln(' come programma mi sento un po'' stupido ...');

      writeln('--------------------------');

   il compilatore non potrebbe in alcun modo capire quali sono le istruzioni

   da ripetere (per quale motivo fermarsi al primo write piuttosto che al

   secondo?); l'aver allineato i comandi write/writeln serve solo a rendere

   piu' leggibile il codice per noi ... ma per il compilatore significa

   comprendere nel ciclo for SOLO LA RIGA SUCCESSIVA:

      for i := 1 to QUANTE_VOLTE do

        write('N. riga: ');

      write(i);

      writeln(' come programma mi sento un po'' stupido ...');

      writeln('--------------------------');

    prova a togliere il begin/end e sperimenta personalmente che il ciclo

    stamperebbe per dieci volte la scritta 'N. riga:' e solo una volta il

    resto !!

    A COSA SERVE ALLORA AVER COMPLICATO IL CODICE ??

    RISPOSTA: AD AVER MAGGIOR CONTROLLO, COME MOSTRATO NELLA 'TERZA SOLUZIONE'*)

(* TERZA SOLUZIONE

   questa volta si 'pretende' che il messaggio che precede il numero di riga

   sia scritto in rosso, il numero di riga in giallo e la scritta in verde

*)

  writeln('TERZA SOLUZIONE');

  for i := 1 to QUANTE_VOLTE do

  begin

    textcolor(RED); write('N. riga: ');(* non va a capo ... *)

    textcolor(YELLOW); write(i);          (* non va a capo ... *)

    textcolor(GREEN); writeln(' come programma mi sento un po'' stupido ...');

  end;

  writeln('--------------------------');

  writeln;

(* commenti: solo costruendo il messaggio una porzione alla volta usando

   il write e' possibile intervallare i comandi per la selezione del colore *)

  writeln('Programma terminato. Premere INVIO per continuare...');

  readln; (* per dare il tempo di leggere il messaggio *)

end.

ITE14 difficoltà: media esercizio 1, pag. 84 del Guidi/Renzoni calcolo delle prime N potenze di 2; deve essere 0<=N<=14

(* DOMANDA: PERCHE' L'AUTORE HA SCELTO PROPRIO FINO A 14 ?? *)

program potenzeDue;

var

  n: integer;    (* il numero inserito dall'utente *)

  pot: integer;  (* le potenze di volta in volta calcolate *)

  i: integer;    (* per i cicli *)

begin

  writeln('CALCOLO POTENZE DI 2');

  writeln;

  (* continuo a chiedere un N finche' soddisfa ... *)

  repeat

    writeln('Inserisci n (0<=n<=14): ');

    readln(n);

    if (n<0) or (n>14) then

      writeln('Valore errato! RIPROVA (0<=n<=14)');

  until (0<=n) and (n<=14);

(* VEDIAMO PRIMA LA SOLUZIONE CON IL REPEAT ...*)

  pot:=1; i:=0;

  writeln('SOLUZIONE CON IL REPEAT');

  repeat

    writeln('La potenza ',i, ' del 2 e'': ',pot);

    pot:=pot*2;

    i:=i+1

  until i>n;

  readln;

(* VEDIAMO POI LA SOLUZIONE CON IL WHILE ...*)

  pot:=1; i:=0;

  writeln('SOLUZIONE CON IL WHILE');

  while i<=n do

  begin

    writeln('La potenza ',i, ' del 2 e'': ',pot);

    pot:=pot*2;

    i:=i+1

  end;

  readln;

(* VEDIAMO INFINE LA SOLUZIONE CON IL FOR ...*)

  writeln('SOLUZIONE CON IL FOR');

  pot:=1;

  for i:=0 to n do

  begin

    writeln('La potenza ',i, ' del 2 e'': ',pot);

    pot:=pot*2

  end;

  readln

end.

ITE15. difficoltà: bassa esercizio 3, pag. 84 del Guidi/Renzoni calcolo media N numeri inseriti dall'utente
program mediaNnumeri;

var

  n: integer;    (* quanti numeri inserire *)

  numero: integer;

  somma: real ;  (* la somma può superare 32767 ... *)

  media: real;

  i: integer;    (* per i cicli *)

begin

  writeln('CALCOLO MEDIA');

  writeln;

  (* continuo a chiedere un N finche' soddisfa ... *)

  repeat

    writeln('quanti numeri vuoi inserire? (N>0): ');

    readln(n);

    if n<=0 then

      writeln('Valore senza senso! RIPROVA (N>0)');

  until n>0;

(* VEDIAMO PRIMA LA SOLUZIONE CON IL REPEAT ...*)

  somma:=0; i:=1;

  writeln('SOLUZIONE CON IL REPEAT');

  repeat

    write('Inserire un numero (',i,') -> ' );

    readln(numero);

    somma:=somma+numero;

    i:=i+1;

  until i>n;

  media:=somma/n;

  writeln('media: ',media:6:2);

  readln;

(* VEDIAMO POI LA SOLUZIONE CON IL WHILE ...*)

  somma:=0; i:=1;

  writeln('SOLUZIONE CON IL WHILE');

  while i<=n do

  begin

    write('Inserire un numero (',i,') -> ' );

    readln(numero);

    somma:=somma+numero;

    i:=i+1;

  end;

  media:=somma/n;

  writeln('media: ',media:6:2);

  readln;

(* VEDIAMO INFINE LA SOLUZIONE CON IL FOR ...*)

  writeln('SOLUZIONE CON IL FOR');

  somma:=0;

  for i:=1 to n do

  begin

    write('Inserire un numero (',i,') -> ' );

    readln(numero);

    somma:=somma+numero;

  end;

  media:=somma/n;

  writeln('media: ',media:6:2);

  readln;

end.

ITE16. difficoltà: media Dato in input un numero intero N, sommare i primi N numeri dispari e

   verificare che tale somma e' uguale al quadrato di N.

program som_disp;

uses newdelay, crt;  

var

  i,quanti: integer;

  numero,somma: real;

begin

  clrscr;

  (* il controllo *sembra* efficace: nel prossimo controllo provate con 35000;

     e poi con 70000: l'errore non viene rilevato; perche' ?? *)

  repeat

    writeln('Quanti numeri dispari vuoi considerare? (da 0 a ',MAXINT,')');

    readln(quanti);

    if (quanti<0) or (quanti>MAXINT) then

      writeln('Errato, ripetere l''inserimento')

  until (quanti>=0) and (quanti<=MAXINT);

  somma:=0;numero:=1; (* 1 e' il primo n. dispari ...*)

  for i:=1 to quanti do

  begin

    somma:=somma+numero;

    numero:=numero+2; (* individuiamo il n. dispari successivo *)

  end;

  writeln('Considerati i primi ',quanti,' numeri dispari');

  writeln('La loro somma e'': ',somma:10:0);

  (* moltiplico per 1.0 per trasformare in real il risultato

     altrimenti sarebbe integer e limitato a 32767 *)

  writeln('Il quadrato di ',quanti, ' e'' ',quanti*1.0*quanti:10:0);

  if somma=quanti*1.0*quanti then

    writeln('Regola verificata!')

  else

    writeln('Regola non verificata!');

 readln;

end.

ITE17. difficoltà: media Un giro turistico e' fatto di N tappe, delle quali si introducono da tastiera

il nome della citta' di arrivo e i km percorsi. Calcolare il percorso totale e il percorso medio delle tappe

program tappe;

uses newdelay, crt;  var

  i,quante_tappe: integer;

  km_tappa,somma_km: real;

  citta: string;

begin

  clrscr;

  (* il controllo *sembra* efficace: nel prossimo controllo provate con 35000;

     e poi con 70000: l'errore non viene rilevato; perche' ?? *)

  repeat

    writeln('Quante sono le tappe? (da 0 a ',MAXINT,')');

    readln(quante_tappe);

    if (quante_tappe<0) or (quante_tappe>MAXINT) then

      writeln('Errato, ripetere l''inserimento')

  until (quante_tappe>=0) and (quante_tappe<=MAXINT);

  somma_km:=0;

  for i:=1 to quante_tappe do

  begin

    writeln('Prossima destinazione ?');

    readln(citta);

    repeat

      writeln('Quanto dista ',citta,' in km?');

      readln(km_tappa);

      if (km_tappa<0) or (km_tappa>MAXINT) then

        writeln('Errato, ripetere l''inserimento')

    until (km_tappa>=0) and (km_tappa<=MAXINT);

    somma_km:=somma_km+km_tappa;

  end;

  writeln('Percorso totale: ',somma_km:5:1,'km');

  writeln('Percorso medio per tappa: ',somma_km/quante_tappe:5:1,'km');

  readln

end.
Sul mio sito (� HYPERLINK "http://www.camuso.it" ��www.camuso.it�) troverete tutti i sorgenti .pas degli esercizi qui presentati 

